
Advanced Excel Formulas: Cheat Sheet
Every formula works on the 10-row sample sheet (Invoice A, Date B, Customer C, Region D, Product E, Qty F, Rate G, Amount H; data in rows 2–11). Versions from Microsoft
Support, checked 16 Oct 2026.

Function Formula Result Excel

XLOOKUP =XLOOKUP("INV-106",A2:A11,H2:H11,"Not found") 22,000 2021+

INDEX + MATCH =INDEX(H2:H11,MATCH("INV-106",A2:A11,0)) 22,000 2016+

Two-way
INDEX/MATCH

=INDEX(A2:H11,MATCH("INV-103",A2:A11,0),MATCH("Qty",A1:H1,0)) 5 2016+

XMATCH (last) =XMATCH("Monitor",E2:E11,0,-1) 10 2021+

SUMIFS =SUMIFS(H2:H11,D2:D11,"North",E2:E11,"Laptop") 2,60,000 2016+

SUMIFS by dates =SUMIFS(H2:H11,D2:D11,"North",B2:B11,">="&DATE(2026,4,1),B2:B11,"
<="&DATE(2026,4,30))

1,68,500 2016+

COUNTIFS =COUNTIFS(H2:H11,">=50000") 4 2016+

MAXIFS =MAXIFS(H2:H11,D2:D11,"East") 28,500 2019+

AVERAGEIFS =AVERAGEIFS(H2:H11,E2:E11,"Monitor") 40,333.33 2016+

SUMPRODUCT =SUMPRODUCT((E2:E11="Laptop")*F2:F11*G2:G11) 3,12,000 2016+

IFS =IFS(H2>=100000,"Large",H2>=25000,"Medium",TRUE,"Small") Large 2019+

SWITCH =SWITCH(D2,"North","N-01","South","S-01","East","E-01","West","W-
01","Unknown")

N-01 2019+

IFERROR =IFERROR(H2/F2,0) 52,000 2016+

FILTER =FILTER(A2:H11,C2:C11="Sharma Traders","No sales") INV-101, INV-104, INV-109 2021+

FILTER (2 conditions) =FILTER(A2:H11,(D2:D11="North")*(H2:H11>=50000)) INV-101, INV-103, INV-109 2021+

UNIQUE + COUNTA =COUNTA(UNIQUE(C2:C11)) 5 2021+

SORT =SORT(A2:H11,8,-1) INV-109 first 2021+

SEQUENCE + DATE =DATE(2026,3+SEQUENCE(12),1) 1-Apr-2026 to 1-Mar-2027 2021+

TEXTJOIN =TEXTJOIN(", ",TRUE,UNIQUE(FILTER(C2:C11,D2:D11="North"))) Sharma Traders, Mehta Stores 2021+

TEXTSPLIT =TEXTSPLIT(J2,"/") (J2 = NEFT/UTR12345/SHARMA TRADERS) NEFT | UTR12345 | SHARMA
TRADERS

2024 / 365

TEXTAFTER (last) =TEXTAFTER(J2,"/",-1) SHARMA TRADERS 2024 / 365

EOMONTH =EOMONTH(B5,0) 30-Apr-2026 2016+

EDATE =EDATE(B5,1) 12-May-2026 2016+

NETWORKDAYS =NETWORKDAYS(DATE(2026,4,1),DATE(2026,4,30)) 22 2016+

TAKE + SORT =TAKE(SORT(A2:H11,8,-1),3) INV-109, INV-101, INV-103 2024 / 365

CHOOSECOLS =CHOOSECOLS(A2:H11,1,3,8) Invoice, Customer, Amount 2024 / 365

VSTACK =VSTACK(April!A2:H8,May!A2:H4) April + May in one list 2024 / 365

LET =LET(amt,H2:H11,north,D2:D11="North",SUM(FILTER(amt,north))) 3,24,500 2021+

LAMBDA (named
NETPRICE)

=LAMBDA(price,disc,price*(1-disc)) then =NETPRICE(G2,10%) 46,800 2024 / 365

GROUPBY =GROUPBY(D2:D11,H2:H11,SUM) E 50,500 | N 3,24,500 | S 65,500 | W
53,000 | Total 4,93,500

365 only

Errors

#N/A: lookup key not found (check spaces, numbers stored as text) · #NAME?: function missing in your version · #SPILL!: clear the cells below · #CALC!: FILTER found nothing, add
the third argument · #REF!: deleted reference.

